

KANSAS GEOLOGICAL SURVEY

PROGRAM PEAP1:

**PLOTTING EQUAL AREA PROJECTIONS
(Using an APPLE II computer)**

**Dave Evans
Kansas Geological Survey
1930 Constant Avenue, Campus West
The University of Kansas
Lawrence, Kansas 66046**

**KGS OPEN-FILE REPORT 85-6
May 23, 1985**

INTRODUCTION

PEAP1 is an interactive program for plotting equal area (Schmidt-net) projections of structural geology data. The program is designed for an APPLE computer interfaced with a Hewlett-Packard 7470A plotter. It is written in BASIC and HP-Graphics language, although a binary version (PEAP1.BIN) exists in order to speed up execution. For a large number of data the execution time for PEAP1 is fairly long, however the program produces high quality, reproducible plots suitable for publication.

Although PEAP1 has some idiosyncrasies, it is easy to use and requires only a modicum of knowledge about computers. In this report, statements enclosed by boxes are prompts produced by PEAP1. This documentation is included in the text file titled PEAP.DOC.

USING PEAP1

GETTING STARTED

To run PEAP1 insert the floppy disk marked 'PEAP1' into disk-drive 1 and turn on the computer. This will automatically load and run the program. Thereafter the program can be run again by typing 'RUN PEAP1', or by turning the computer off and back on.

INPUT

- <1> PLOT POLES (Pi-diagram)
- <2> PLOT PLANES (Beta-diagram)
- <3> PLOT % OF POINTS (For contouring)
- <4> PLOT ENTIRE NET (Schmidt net)

From this prompt the user specifies the type of plot desired.

ENTER PLOT TITLE (30 characters or less)

A title identifying the plot will be printed at the lower left hand corner of the plot. Enter the title here. A response of <cr> will cause no title to be printed. A response of '=' implies that a plot has already been made and the next plot will be superimposed on the existing plot.

ENTER DIAMETER OF PLOT (in cm)

PEAP1 will make a plot of any diameter between 1 cm and 15

cm. A response of <cr> will produce a plot 10 cm (the default value) in diameter. (Note: the value provided for diameter will only be approximate.)

ENTER ANGULAR INCREMENT FOR PLOT (in degrees)

The HP 7470a plotter plots only straight-line segments. From this prompt, then, the user specifies the angular distance between end points of these segments. If this value is too small (say 1 degree) it will take a very long time to produce the plot and the lines will be fuzzy. If this value is too big (say > 10 degrees) the circumference of the plot will be obviously polygonal. A value of 5 or 10 degrees is recommended. The default value is 10 degrees.

ENTER PEN VELOCITY (in cm)

Pen velocity controls the quality of the lines made by the plotter. If the pen speed is too slow (around 10 cm/sec), the plot will take a long time to produce and the lines will be fuzzy. If the pen speed is too fast the lines will be too thin. With a new pen and good paper, a velocity of 25 to 38 cm/sec is recommended. The default value is 38 cm/sec.

ENTER NAME OF FILE CONTAINING DATA

In many circumstances you may want to plot data that was input in an earlier run. If the data exist on a disk file, the name of that file is specified here. A response of 'NONE' or <cr> means that new data will be input. A response of '=' means that a plot has already been made and the next plot will use data input previously in the current run.

If a file name was specified, you (and the plotter) are ready to plot. However, if data are to be input you will get the following prompt.

Input Options

- <1> ENTER TREND AND PLUNGE OF LINES
- <2> ENTER STRIKE AND DIP OF PLANES

From this prompt you specify the kind of data that will be input. This also initiates the appropriate set of input prompts.

ENTER TREND OF LINE (in degrees)

The trend must be input as an azimuth between 0 and 360 degrees. Negative values between 0 and -90 degrees are also allowed.

ENTER STRIKE OF PLANE (in degrees)

(Format is the same as for trend)

ENTER PLUNGE OF LINE (in degrees)

The plunge must be input as a value between 0 and 90 degrees followed by either an 'E' or a 'W' (NW, SE, N, etc. are not allowed). A plunge due south must be input as 'E' and a plunge due north must be input as 'W'.

ENTER DIP OF PLANE (in degrees)

(Format is the same as for trend)

ENTER SYMBOL TO BE PLOTTED

If a Pi-diagram has been specified, different symbols (any single symbol except '!') can be plotted. The default symbol is '.'. .

To terminate input, enter a strike (or trend) of '0' and a dip (or plunge) of '0'.

After all of the data have been input, they are automatically listed for inspection. You will note that strikes (or trends) listed may be 180 degrees less than the value you entered. This is due to the way the program determines which way the planes dip. When the listing is complete, any values can be changed by responding to the prompt

ENTER NUMBER OF DATA TO BE CHANGED (<cr> to continue)

After editing, you can save the data on a disk file.

SAVE DATA TO DISK (Y/N) ?

ENTER SAVE FILE NAME

A file name can be any combination of letters and numbers up to 32 characters. The file name must start with a letter (not a number).

Now you (and the plotter) are ready to plot.

LOAD PAPER AND PEN. <cr> TO PLOT.

A pen must be loaded in the left stall.

Once the plotting is complete, the user can terminate the program or make another plot by responding to the prompt

ENTER 'C' TO CONTINUE
'E' TO END


Dave Evans

Oct. 23, 1983


REFERENCES

- Duncan, Andrew D. (1981). A review of Cartesian coordinate construction from a sphere, for generation of two dimensional geological net projections, COMPUTERS & GEOSCIENCES, vol 7, no 4, pp367-385
- Hobbs, B.E., W.D.Means, and P.F.Williams (1976). AN OUTLINE OF STRUCTURAL GEOLOGY, John Wiley and Sons, New York, 571p.


APPENDIX I**SAMPLE OUTPUT FROM PEAP1**




SAMPLE RUN NUMBER 1



SAMPLE RUN NUMBER 1A



TEST RUN NUMBER 6



APPENDIX II

LISTING OF PEAP1

```

90 PRINT CHR$(4);"PR#3": HOME : PRINT H$
100 REM !-----!
110 REM !---          PROGRAM PEAP1          ---!
120 REM !-----!
130 DIM PL(300),TR(300),SYM$(300),XX(300),YY(300),AZ(300),DP(300)
132 DIM X(300),Y(300),Q$(300)
135 DIM GX(20),GY(20)
140 PI = 3.14159265
150 RAD = PI / 180
155 SR2 = SQR (2)
156 X$ = CHR$(3)
158 H$ = CHR$(12)
159 D$ = CHR$(4)
162 DEF FN ASN(X) = ATN (X / SQR (- X * X + 1))
164 REM
166 REM
170 GOSUB 1000: REM >>>> CALL DATAIN <<<<
175 IF FLAG(3) = 1 THEN PR# 2: GOTO 190
176 HOME : PRINT CHR$(12)
177 VTAB 12: PRINT "LOAD PAPER AND PEN. <cr> TO PLOT."
178 INPUT W$: PRINT : PRINT : IF LEN (W$) < > 0 THEN GOTO 176
180 GOSUB 2000: REM >>>> CALL SPHPLOT <<<<
188 REM
190 REM >>>> ON FLAG(1), CALL PTPLOT, CALL PLNLOT, CALL NETPLOT <<<<
200 REM
210 ON FLAG(1) GOSUB 5000,6000,8000,7000
215 PR# 3
220 HOME : PRINT CHR$(12): HOME : PRINT CHR$(12)
230 VTAB 12: PRINT "ENTER 'C' TO CONTINUE,"
240 INPUT "      'E' TO END      ";EC$
250 IF EC$ = "C" THEN GOTO 170
260 IF EC$ < > "E" THEN GOTO 220
270 PR# 2
280 PRINT "SP0;DF;"
290 PR# 3
300 HOME : PRINT CHR$(12)
302 VTAB 12: PRINT "<<<<<<<<<< HIT 'RESET' BEFORE RUNNING AGAIN >>>>>>>>>"
310 END
500 REM
510 REM
520 REM
530 REM
540 REM
550 REM
1000 REM !-----!
1007 REM !---          SUBROUTINE DATAIN          ---!
1014 REM !-----!
1021 HOME : PRINT H$
1028 PRINT TAB(34);"Plot Options"
1035 PRINT : PRINT : PRINT TAB(20);"<1> PLOT POLES (Pi-diagram)"
1042 PRINT : PRINT TAB(20);"<2> PLOT PLANES (Beta-diagrams)"
1049 PRINT : PRINT TAB(20);"<3> PLOT % OF POINTS (For contours)"
1056 PRINT : PRINT TAB(20);"<4> PLOT ENTIRE NET (Schmidt net)"
1063 PRINT : INPUT FLAG(1)
1070 IF FLAG(1) < 1 OR FLAG(1) > 4 THEN GOTO 1021
1077 HOME : PRINT H$
1084 PRINT TAB(34);"Plot Format"
1091 PRINT : PRINT : PRINT "ENTER PLOT TITLE (30 Characters or less)"

```

```

1098 INPUT " : ";A$
1105 IF A$ = "=" THEN FLAG(3) = 1: GOTO 1175
1112 TITLE$ = A$
1119 PRINT : PRINT : PRINT "ENTER DIAMETER OF PLOT (in cm) "
1120 INPUT "      (default = 10)      ";CR$
1123 IF LEN (CR$) = 0 THEN CR$ = "10"
1124 CR = VAL (CR$)
1126 IF CR > 15 THEN CR = 15
1133 IF CR < 1 THEN CR = 1
1140 CR = CR / 2
1147 PRINT : PRINT : PRINT "ENTER ANGULAR INCREMENT FOR PLOT (in degrees) "
1154 INPUT "      (default = 10)      ";VSTP$
1161 IF LEN (VSTP$) = 0 THEN VSTP$ = "10"
1168 VSTP = VAL (VSTP$)
1169 PRINT : PRINT : PRINT "ENTER PEN VELOCITY (in cm/sec) "
1170 INPUT "      (default = 38)      ";PV$
1171 IF LEN (PV$) = 0 THEN PV$ = "38.1"
1172 PV = VAL (PV$)
1175 PRINT : HOME : PRINT H$
1182 IF FLAG(1) = 4 THEN RETURN
1189 OP$(1) = "Plot Poles"
1196 OP$(2) = "Plot Planes"
1203 OP$(3) = "Plot Point Z"
1210 PRINT TAB( 32);OP$(FLAG(1))
1217 HOME : PRINT H$: PRINT TAB( 33);OP$(FLAG(1))
1224 PRINT : PRINT : PRINT TAB( 20);"ENTER NAME OF FILE CONTAINING DATA": INPUT F1$
1225 IF F1$ = "=" THEN RETURN
1231 IF LEN (F1$) = 0 THEN F1$ = "NONE"
1238 IF F1$ = "NONE" THEN GOTO 1315
1245 IF FLAG(3) = 1 THEN RETURN
1252 PRINT D$;"OPEN ";F1$
1259 PRINT D$;"READ ";F1$
1266 FOR I = 1 TO 300
1273 INPUT AZM(I),DP(I),TR(I),PL(I),SYM$(I)
1280 IF SYM$(I) = "!" THEN GOTO 1294
1287 NEXT I
1294 PRINT D$;"CLOSE ";F1$
1301 K = I - 1
1308 GOTO 1749
1315 HOME : PRINT H$: PRINT TAB( 31);"Input Options"
1322 PRINT : PRINT : PRINT TAB( 20);"<1> ENTER TREND AND PLUNGE OF LINES"
1329 PRINT : PRINT TAB( 20);"<2> ENTER STRIKE AND DIP OF PLANES"
1336 INPUT FLAG(2)
1343 IF FLAG(2) > 2 THEN GOTO 1210
1350 IF FLAG(2) = 2 THEN GOTO 1553
1357 FOR I = 1 TO 300
1364 HOME : PRINT H$
1371 PRINT TAB( 32);OP$(FLAG(1))
1378 PRINT TAB( 30);"Input Line Data"
1385 VTAB 7
1392 PRINT : INPUT "ENTER TREND OF LINE (in degrees) : ";TR(I)
1394 AZM(I) = TR(I) + 90
1399 IF AZM(I) > 360 THEN AZM(I) = AZM(I) - 360
1406 PRINT : INPUT "ENTER PLUNGE OF LINE (in degrees) : ";PL$
1408 Q$(I) = PL$
1413 PL(I) = VAL (PL$)
1415 IF TR(I) = 0 AND PL(I) = 0 THEN GOTO 1462
1420 PL$ = RIGHT$ (PL$,1)
1427 IF AZM(I) > 180 THEN AZM(I) = AZM(I) - 180

```

```

1428 IF AZM(I) > 90 THEN AZM(I) = AZM(I) - 180
1429 IF AZM(I) < - 90 THEN AZM(I) = AZM(I) + 180
1430 IF PL$ = "E" THEN AZM(I) = AZM(I) - 180
1431 TR(I) = AZM(I) - 90
1434 DP(I) = PL(I) + 90
1435 SYM$(I) = "."
1441 IF FLAG(I) = 1 THEN PRINT : INPUT "ENTER SYMBOL TO BE PLOTTED : ";SYM$(I)

1443 IF LEN (SYM$(I)) = 0 THEN SYM$(I) = "."
1448 IF SYM$(I) = "!" THEN GOTO 1462
1455 NEXT I
1462 K = I - 1
1464 FLAG(6) = 1: GOSUB 9000: REM >>>> CALL DATAEDIT <<<<<
1469 GOTO 1749
1476 PRINT : PRINT TAB( 32);OP$(FLAG(1))
1483 PRINT : INPUT "SAVE DATA TO DISK (Y/N) ? ";A$
1490 IF A$ = "N" THEN GOTO 1749
1497 PRINT : INPUT "ENTER SAVE FILE NAME : ";F2$
1504 PRINT D$;"OPEN ";F2$
1511 PRINT D$;"WRITE ";F2$
1518 FOR I = 1 TO K
1525 PRINT AZM(I);", ";DP(I);", ";TR(I);", ";PL(I);", ";SYM$(I)
1532 NEXT I
1539 PRINT 0;";";0;";";0;";";0;";";0;";";!"
1546 PRINT D$;"CLOSE ";F2$
1553 FOR I = 1 TO 500
1560 HOME : PRINT H$
1567 PRINT TAB( 32);OP$(FLAG(1))
1574 PRINT TAB( 30);"Input Plane Data"
1581 VTAB 7
1588 INPUT "ENTER STRIKE OF PLANE (in degrees) : ";AZM(I)
1602 PRINT : INPUT "ENTER DIP OF PLANE (in degrees) : ";DP$
1605 Q$(I) = DP$
1609 DP(I) = VAL (DP$)
1616 PL(I) = DP(I) + 90
1623 DP$ = RIGHT$ (DP$,1)
1624 IF AZM(I) > 180 THEN AZM(I) = AZM(I) - 180
1625 IF AZM(I) > 90 THEN AZM(I) = AZM(I) - 180
1626 IF AZM(I) < - 90 THEN AZM(I) = AZM(I) + 180
1630 IF DP$ = "E" THEN AZM(I) = AZM(I) - 180
1632 TR(I) = AZM(I) - 90
1637 SYM$(I) = "."
1644 IF AZM(I) = 0 AND DP(I) = 0 THEN GOTO 1658
1645 IF FLAG(I) = 1 THEN PRINT : INPUT "ENTER SYMBOL TO BE PLOTTED : ";SYM$(I)

1646 IF LEN (SYM$(I)) = 0 THEN SYM$(I) = "."
1651 NEXT I
1658 K = I - 1
1660 FLAG(6) = 2: GOSUB 9000: REM >>>> CALL DATAEDIT <<<<<
1665 HOME : PRINT H$
1672 PRINT TAB( 33);OP$(FLAG(1))
1679 PRINT : INPUT "SAVE DATA TO DISK (Y/N) ? ";A$
1686 IF A$ = "N" THEN GOTO 1749
1693 PRINT : INPUT "ENTER SAVE FILE NAME : ";F2$
1700 PRINT D$;"OPEN ";F2$
1707 PRINT D$;"WRITE ";F2$
1714 FOR I = 1 TO K
1721 PRINT AZM(I);", ";DP(I);", ";TR(I);", ";PL(I);", ";SYM$(I)
1728 NEXT I
1735 PRINT 0;";";0;";";0;";";0;";";0;";";!"

```

```

1742 PRINT D$;"CLOSE ";F2$
1749 FOR I = 1 TO K
1756 AZM(I) = AZM(I) * RAD
1763 DP(I) = DP(I) * RAD
1770 NEXT I
1777 RETURN
2000 REM !-----!
2010 REM !---          SUBROUTINE SPHPLOT          ---!
2020 REM !-----!
2030 REM
2040 REM
2050 PR# 2
2060 PRINT "IN;SP1;"
2070 PRINT "IPS150,3825,8975,7650;"
2080 PRINT "SCO,9.5625,0,9.5625;"
2085 PRINT "VS";PV;"
2090 REM
2100 REM --- PLOT OUTSIDE OF SPHERE
2110 REM
2120 FOR J = 0 TO 360 STEP VSTP
2130 T = J * RAD
2140 X = CR * COS (T)
2150 Y = CR * SIN (T)
2160 IF ABS (X) < 0.01 THEN X = 0
2170 IF ABS (Y) < 0.01 THEN Y = 0
2180 PRINT "PA",X,Y,";PD;"
2190 IF J = 0 THEN PRINT "TL2;YT;"
2200 IF J = 90 THEN PRINT "XT;"
2210 IF J = 180 THEN PRINT "TL-2;YT;"
2220 IF J = 270 THEN PRINT "XT;"
2230 NEXT J
2240 PRINT "PU;"
2250 REM
2260 REM --- LABEL SPHERE
2270 REM
2280 PRINT "SI.3,.3;DIO,-1;"
2290 T = 0
2300 X = CR * COS (T) + 0.75:Y = 0
2310 PRINT "SMN;"
2320 PRINT "PA",X,Y,";PU;"
2330 T = PI / 2
2340 X = 0:Y = CR * SIN (T) + 0.75
2350 PRINT "SMW;"
2360 PRINT "PA",X,Y,";PU;"
2370 T = PI
2380 X = CR * COS (T) - 0.75:Y = 0
2390 PRINT "SMS;"
2400 PRINT "PA",X,Y,";PU;"
2410 T = 3 * PI / 2
2420 X = 0:Y = CR * SIN (T) - 0.75
2430 PRINT "SME;"
2440 PRINT "PA",X,Y,";PU;"
2450 PRINT "SM;"
2454 PRINT "PA-10,8.5;"
2455 PRINT "SI.25,.3;"
2456 PRINT "LB";TITLE$;X$
2458 PRINT "PU;PA-10,9.5;"
2459 FOR I = 1 TO 400: NEXT I
2460 RETURN
2500 REM

```

```

2510 REM
2520 REM
2530 REM
2540 REM
2550 REM
5000 REM !-----!
5010 REM !---          SUBROUTINE PTPLOT          ---!
5020 REM !-----!
5030 REM
5040 FOR I = 1 TO K
5050 PRINT "SI.2 ,.2 ;PU;"
5055 RHO = PL(I) * RAD
5057 PHI = TR(I) * RAD
5060 EL = CR * SR2 * SIN (PI / 4 - RHO / 2)
5070 X = EL * COS (PHI)
5080 Y = - EL * SIN (PHI)
5090 IF ABS (X) < 0.01 THEN X = 0
5100 IF ABS (Y) < 0.01 THEN Y = 0
5102 X(I) = X;Y(I) = Y
5104 IF FLAG(1) = 3 THEN GOTO 5140
5110 PRINT "SM";SYM$(I);";"
5120 PRINT "PA",X,Y,";"
5140 NEXT I
5145 PRINT "SM;"
5150 RETURN
5300 REM
5310 REM
5320 REM
5330 REM
5340 REM
5350 REM
6000 REM !-----!
6010 REM !---          SUBROUTINE PLNPLLOT          ---!
6020 REM !-----!
6021 FOR I = 1 TO K
6022 CD = COS ( - DP(I))
6024 SD = SIN ( - DP(I))
6025 DLTA = - DP(I)
6026 ST = AZM(I)
6030 FOR P = 0 TO 180 STEP VSTP
6040 PSI = P * RAD
6050 XP = COS (PSI)
6060 YP = - SIN (PSI) * CD
6070 Z = - SIN (PSI) * SD
6080 IF XP = 0 THEN XP = 1E - 08
6090 PHI = ATN (YP / XP)
6100 RHO = FN ASN(Z)
6110 RHO = ABS (RHO)
6120 EL = CR * SR2 * SIN (PI / 4 - RHO / 2)
6130 X = EL * COS (PHI + ST)
6140 Y = - EL * SIN (PHI + ST)
6150 IF P > 90 THEN X = - X;Y = - Y
6160 IF ABS (X) < 0.01 THEN X = 0
6170 IF ABS (Y) < 0.01 THEN Y = 0
6180 PRINT "PA",X,Y,";PD;"
6190 NEXT P
6200 PRINT "PU;"
6210 IF FLAG(1) < > 4 THEN GOTO 6245
6215 KOUNT = KOUNT + 1
6220 IF KOUNT > = 2 THEN RETURN

```

```

6230 DLTA = PI / 2 - DLTA * COS (S0)
6232 CD = COS (DLTA)
6234 SD = SIN (DLTA)
6240 GOTO 6030
6245 NEXT I
6250 RETURN
6260 END
6300 REM
6310 REM
6320 REM
6330 REM
6340 REM
6350 REM
7000 REM !-----!
7010 REM !---          SUBROUTINE NETPLOT          ---!
7020 REM !-----!
7030 REM
7040 REM
7050 REM >>>>> PLOT GREAT CIRCLES <<<<<<
7060 REM
7070 ST(1) = 0
7080 ST(2) = 0
7090 FOR D = 10 TO 90 STEP 10
7095 KOUNT = 0
7100 DLTA = - D * RAD
7105 CD = COS (DLTA)
7106 SD = SIN (DLTA)
7110 GOSUB 6030: REM >>>>> CALL PLNPLT <<<<<<
7120 NEXT D
7130 REM
7140 REM >>>>> PLOT SMALL CIRCLES <<<<<<
7150 REM
7160 FOR P = 10 TO 180 STEP VSTP
7170 PSI = P * RAD
7180 SP = SIN (PSI)
7190 CP = COS (PSI)
7200 FOR D = 0 TO 180 STEP 10
7210 DLTA = - D * RAD
7220 XP = CP
7230 YP = - SP * COS (DLTA)
7240 Z = - SP * SIN (DLTA)
7250 IF XP = 0 THEN XP = 1E - 08
7260 PHI = ATN (YP / XP)
7270 RHO = FN ASN(Z)
7280 RHO = ABS (RHO)
7290 EL = CR * SR2 * SIN (PI / 4 - RHO / 2)
7300 X = EL * COS (PHI)
7310 Y = - EL * SIN (PHI)
7320 IF P > 90 THEN X = - X:Y = - Y
7330 IF ABS (X) < 0.01 THEN X = 0
7340 IF ABS (Y) < 0.01 THEN Y = 0
7350 PRINT "PA",X,Y,";PD;"
7360 NEXT D
7370 PRINT "PU;"
7380 NEXT P
7390 RETURN
7400 END
7500 REM
7510 REM
7520 REM

```

```

7530 REM
7540 REM
7550 REM
8000 REM !-----!
8010 REM !---          SUBROUTINE PCTPLOT          ---!
8020 REM !-----!
8030 REM
8035 GOSUB 5000: REM >>>> CALL PTPLT <<<<
8040 REM
8050 REM --- INITIALIZE GRID FOR COUNTING POINTS
8060 FOR I = 0 TO 20
8070 GX(I) = CR - (0.1 * I * CR)
8080 GY(I) = CR - (0.1 * I * CR)
8090 NEXT I
8100 LR = CR / 10
8120 FOR I = 0 TO 20
8130 X0 = GX(I)
8140 FOR J = 0 TO 20
8141 PRINT "SI.03,.03;SM.;PU;"
8150 Y0 = GY(J)
8160 DIST = SQR (X0 * X0 + Y0 * Y0)
8170 IF DIST > CR THEN GOTO 8560
8180 PRINT "PA",X0,Y0,";PU;"
8190 REM --- FIND OUT IF (X0,Y0) IS A BORDER POINT
8200 IF (DIST + LR) > CR THEN GOTO 8400
8210 REM --- IF NOT A BORDER POINT, COUNT THE POINTS W/IN LR
8220 NP = 0
8230 FOR IJ = 1 TO K
8240 Q = SQR ((X(IJ) - X0) ^ 2 + (Y(IJ) - Y0) ^ 2)
8250 IF Q < = LR THEN NP = NP + 1
8260 NEXT IJ
8270 GOTO 8550
8400 REM --- IF (X0,Y0) IS A BORDER POINT...
8401 NP = 0
8403 I1 = 20 - I;J1 = 20 - J
8404 IF I < 10 THEN I2 = I1 + 1
8405 IF J < 10 THEN J2 = J1 + 1
8406 IF I > 10 THEN I2 = I1 - 1
8407 IF J > 10 THEN J2 = J1 - 1
8410 X1 = GX(I2);Y1 = GY(J2)
8420 FOR IJ = 1 TO K
8430 Q0 = SQR ((X(IJ) - X0) ^ 2 + (Y(IJ) - Y0) ^ 2)
8440 Q1 = SQR ((X(IJ) - X1) ^ 2 + (Y(IJ) - Y1) ^ 2)
8450 IF Q0 < = LR OR Q1 < = LR THEN NP = NP + 1
8460 NEXT IJ
8550 NP = INT ((NP / K) * 100 + .5)
8551 IF NP > 0 THEN PRINT "SI.15,.15;LB";NP;X%;
8560 NEXT J
8570 NEXT I
9000 REM !-----!
9010 REM !---          SUBROUTINE DATAEDIT          ---!
9020 REM !-----!
9030 REM
9040 REM
9050 HOME : PRINT H$
9060 IF FLAG(6) = 2 THEN GOTO 9500
9070 FOR I = 1 TO K
9080 PRINT I,TR(I),Q$(I)
9090 IF INT (I / 20) < > I / 20 THEN GOTO 9100
9092 GET B$: PRINT

```

```

9100 NEXT I
9110 PRINT : PRINT : INPUT "ENTER NUMBER OF DATA TO BE CHANGED (<cr> TO CONTINU
E) : ";A$
9120 IF LEN (A$) = 0 THEN RETURN
9130 J = VAL (A$)
9140 HOME : PRINT H$
9150 PRINT TAB( 36);"Old Data"
9160 PRINT : PRINT TAB( 20);"TREND(";J;") = ";TR(J)
9170 PRINT : PRINT TAB( 20);"PLUNGE(";J;") = ";Q$(J)
9180 PRINT : PRINT : PRINT TAB( 33);"Enter New Data"
9190 PRINT : PRINT TAB( 20);: INPUT "ENTER NEW TREND : ";TR(J)
9200 AZM(J) = TR(J) + 90
9202 IF AZM(J) > 360 THEN AZM(J) = AZM(J) - 360
9210 PRINT : PRINT TAB( 20);: INPUT "ENTER NEW PLUNGE : ";Q$(J)
9230 PL$ = Q$(J)
9240 PL(J) = VAL (PL$)
9250 PL$ = RIGHT$ (PL$,1)
9252 IF AZM(J) > 180 THEN AZM(J) = AZM(J) - 180
9254 IF AZM(J) > 90 THEN AZM(J) = AZM(J) - 180
9255 IF AZM(J) < - 90 THEN AZM(J) = AZM(J) + 180
9260 IF PL$ = "E" THEN AZM(J) = AZM(J) - 180
9262 TR(J) = AZM(J) - 90
9270 DP(J) = PL(J) + 90
9280 PRINT : PRINT : INPUT "LIST AGAIN (Y/N) ? ";A$
9290 IF A$ = "Y" THEN GOTO 9050
9300 IF A$ = "Y" THEN RETURN
9310 GOTO 9280
9500 FOR I = 1 TO K
9510 PRINT I,AZM(I),Q$(I)
9520 IF INT (I / 20) < > I / 20 THEN GOTO 9530
9525 GET B$: PRINT
9530 NEXT I
9540 PRINT : PRINT : INPUT "ENTER NUMBER OF DATA TO BE CHANGED (<cr> TO CONTINU
E) : ";A$
9550 IF LEN (A$) = 0 THEN RETURN
9555 J = VAL (A$)
9560 HOME : PRINT H$
9570 PRINT TAB( 36);"Old Data"
9580 PRINT : PRINT TAB( 20);"AZIMUTH(";J;") = ";AZM(J)
9590 PRINT : PRINT TAB( 20);"DIP(";J;") = ";Q$(J)
9600 PRINT : PRINT : PRINT TAB( 33);"Enter New Data"
9610 PRINT : PRINT TAB( 20);: INPUT "ENTER NEW AZMIUTH : ";AZM(J)
9620 PRINT : PRINT TAB( 20);: INPUT "ENTER NEW DIP : ";Q$(J)
9630 DP$ = Q$(J)
9640 DP(J) = VAL (DP$)
9650 DP$ = RIGHT$ (DP$,1)
9652 IF AZM(J) > 180 THEN AZM(J) = AZM(J) - 180
9654 IF AZM(J) > 90 THEN AZM(J) = AZM(J) - 180
9655 IF AZM(J) < - 90 THEN AZM(J) = AZM(J) + 180
9660 IF DP$ = "E" THEN AZM(J) = AZM(J) - 180
9662 TR(J) = AZM(J) - 90
9670 PL(J) = DP(J) + 90
9680 PRINT : PRINT : INPUT "LIST AGAIN (Y/N) ? ";A$
9690 IF A$ = "Y" THEN GOTO 9050
9700 IF A$ = "N" THEN RETURN
9710 GOTO 9680
9720 RETURN

```